반도체 소자 설계 7조



16010783 정유종
최창희
이택진
김현성

목차

1 설계에 사용 된 기술 3가지

① Texturing (표면 구조화)

실리콘의 표면반사도 반사도 40%

Texturing 공정을 이용하여 15%↓

1 설계에 사용 된 기술 3가지

② 이중 반사 방지막



실리콘 태양전지의 반사방지막으로 는 수소를 포함한 SiNx(실리콘 질화막)이 가 장 많이 사용

단층 반사방지막 (SLARC)으로 사용되는 SiNx(실리콘 질화 막)만으로는 400nm-1000nm의 일정 파장 범위를 제외한 나머지 파장에서 반사도가 급격히 증가

이에 태양광 흡수 스펙트럼 에서 더 넓은 영역의 입사광을 사용할 수 있는 이중 반사 방지막(DLARC)을 사용

1 설계에 사용 된 기술 3가지

③ Perc



Perc 기술에서 새로운 반사층은 장파장의 태양빛을 전지안으로 반사

뒷면으로 빠져나가는 장파장의 태양광으로 인한 태양전지의 온도상승에 따른 효율저하↓

전지내에 생성된 전공의 접촉면을 줄여 전공의 로스↓

2 태양전지 Deckbuild Code

① athena code

```
go athena

set thickness=150
set texturing=3

line x loc=0 spac=2
line x loc=420 spac=2
line y loc=0 spac=3
line y loc=3  spac=3
line y loc=90 spac=40
line y loc=$"thickness" spac=3

#p-type기판
init silicon c.boron=3.3e16 two.d
etch silicon start x=0 y=$"texturing"*0.803
etch cont x=0 y=0
etch cont x=420 y=0
etch done x=420 y=$"texturing"*0.803

#SiNx
depo nitride thickness=0.08 div=1
etch nitride start x=100 y=0
etch cont x=100 y=3
etch cont x=110 y=3
etch done x=110 y=0
etch nitride start x=310 y=0
etch cont x=310 y=3
etch cont x=320 y=3
etch done x=320 y=0

#전면 전극 형성
depo material=silver thickness=10
etch material=silver left p1.x=100
etch material=silver right p1.x=320
etch material=silver start x=110 y=-10.1
etch cont x=110 y=$"texturing"+0.1
etch cont x=310 y=$"texturing"+0.1
etch done x=310 y=-10.1

#후면 BSF(Back Surface Field) 및 후면 전극 형성
structure flip
diffuse time=120 temp=1100 c.boron=1e21
depo material=aluminum thickness=5 div=1
structure flip

#전극 정의
electrode name=cathode x=105 y=-1
electrode name=cathode x=315 y=-1
electrode name=anode x=210 y=$"thickness"+1

structure outf=basic.str

tonyplot basic.str
```

2 태양전지 Deckbuild Code

② atlas code

```
go atlas

mesh infile=basic.str

set "area(um2)"=420
set "area(cm2)"=4.2e-6

#emitter 형성
doping gauss n.type conc=5e18 peak=$"texturing"*0.803 char=0.45

#물질 특성, beam, model 정의
material material=silicon taun0=3e-4 taup0=3e-4
beam num=1 x.origin=210 y.origin=-20 angle=90.0 front.refl back.refl reflect=3 metal.refl aml.5
models conmob fldmob consrh


solve init
log outf=basic.log
solve b1=1 vanode=0 vstep=0.01 vfinal=0.7 name=anode
log off

tonyplot basic.log


#parameter 추출
extract init infile="basic.log"
extract name="Isc(A)" max(i."cathode")
extract name="Jsc(A/cm2)" $"Isc(A)"/$"area(cm2)"
extract name="Voc(V)" x.val from curve(v."anode", i."cathode") where y.val=0.0
extract name="Power" curve(v."anode", (v."anode"*i."cathode")) outf="basic.dat"
extract name="Pmax(W)" max(curve(v."anode", (v."anode"*i."cathode")))
extract name="Fill_factor" ($"Pmax(W)"/($"Isc(A)"*$"Voc(V)"))
extract name="Efficiency(%)" ($"Jsc(A/cm2)"*$"Voc(V)"*$"Fill_factor"*100/0.1)
```

2 태양전지 Deckbuild Code

③ 구조



3 3가지 기술을 적용한 Deckbuild Code

① atlas code

```
go athena

set thickness=150
set texturing=3

line x loc=0 spac=2
line x loc=420 spac=2
line y loc=0 spac=3
line y loc=3 spac=3
line y loc=90 spac=40
line y loc=$"thickness" spac=2

##p-type기판
init silicon c.boron=3.3e16 two.d
```

Texturing

```
#texturing
etch silicon start x=0 y=0
etch cont x=0 y=$"texturing"*0.803
etch cont x=114 y=$"texturing"*0.803
etch done x=114 y=0
etch silicon start x=306 y=0
etch cont x=306 y=$"texturing"*0.803
etch cont x=420 y=$"texturing"*0.803
etch done x=420 y=0
foreach val (114 to 298 step 48)
etch silicon start x=val y=0
etch cont x=val y=$"texturing"*0.803
etch cont x=val+1 y=$"texturing"*0.803
etch done x=val+4 y=0
etch silicon start x=val+4 y=0
etch cont x=val+7 y=$"texturing"*0.803
etch cont x=val+48 y=$"texturing"*0.803
etch done x=val+48 y=0
end
```

```
##전면 전극 형성
depo material=silver thickness=10
etch material=silver left p1.x=100
etch material=silver right p1.x=320
etch material=silver start x=110 y=-10.1
etch cont x=110 y=($"texturing"*0.803)+0.1
etch cont x=310 y=($"texturing"*0.803)+0.1
etch done x=310 y=-10.1
```

Perc

```
##perc
structure flip
#BSF
depo material=alumin thickness=1 divi=1
etch material=alumin start x=106 y=-($"thickness"+1.1)
etch cont x=106 y=-($"thickness"-4.1)
etch cont x=114 y=-($"thickness"-4.1)
etch done x=114 y=-($"thickness"+1.1)
etch material=alumin start x=306 y=-($"thickness"+1.1)
etch cont x=306 y=-($"thickness"-4.1)
etch cont x=314 y=-($"thickness"-4.1)
etch done x=314 y=-($"thickness"+1.1)
diffuse time=120 temp=1100 c.boron=2e21
etch material=alumin all

#passivation layer
depo oxide thickness=1 div=1
etch oxide start x=105 y=-($"thickness"+1.1)
etch cont x=105 y=-($"thickness"-1.1)
etch cont x=115 y=-($"thickness"-1.1)
etch done x=115 y=-($"thickness"+1.1)
etch oxide start x=305 y=-($"thickness"+1.1)
etch cont x=305 y=-($"thickness"-1.1)
etch cont x=315 y=-($"thickness"-1.1)
etch done x=315 y=-($"thickness"+1.1)
```

```
#capping layer
depo nitride thickness=1 div=1
etch nitride start x=105 y=-($"thickness"+2.1)
etch cont x=105 y=-($"thickness"-0.1)
etch cont x=115 y=-($"thickness"-0.1)
etch done x=115 y=-($"thickness"+2.1)
etch nitride start x=305 y=-($"thickness"+2.1)
etch cont x=305 y=-($"thickness"-0.1)
etch cont x=315 y=-($"thickness"-0.1)
etch done x=315 y=-($"thickness"+2.1)

##후면 전극(anode 형성)
depo alumin thickness=15 div=1
etch alumin start x=0 y=-170
etch cont x=0 y=-157
etch cont x=420 y=-157
etch done x=420 y=-170

structure flip

##전극 정의
electrode name=cathode x=105 y=-1
electrode name=cathode x=315 y=-1
electrode name=anode x=210 y=$"thickness"+3

structure outf=solution.str

#tonyplot solution.str
```

3 3가지 기술을 적용한 Deckbuild Code

② atlas code

```
go atlas

mesh infile=solution.str

set "area(um2)"=420
set "area(cm2)"=4.2e-6

##emitter 형성
doping gauss n.type conc=5e18 peak=$"texturing"*0.803 char=0.45

##물질 특성, beam, model 정의
material material=silicon taun0=3e-4 taup0=3e-4
beam num=1 x.origin=210 y.origin=-20 angle=90.0 front.refl back.refl reflect=3 metal.refl aml.5
models conmob fldmob consrh

##이중반사방지막
interface optical ar.index=2.22 ar.thick=0.08 p1.x=0 p1.y=$"texturing"*0.803 p2.x=420 p2.y=$"texturing"*0.803
##Al2O3
interface optical ar.index=1.72 ar.thick=0.08 p1.x=0 p1.y=$"texturing"*0.803-0.08 p2.x=420 p2.y=$"texturing"*0.803-0.08

solve init
log outf=solution.log
solve b1=1 vanode=0 vstep=0.01 vfinal=0.7 name=anode
log off

tonyplot solution.log


##parameter 추출
extract init infile="solution.log"
extract name="Isc(A)" max(i."cathode")
extract name="Jsc(A/cm2)" $"Isc(A)"/$"area(cm2)"
extract name="Voc(V)" x.val from curve(v."anode", i."cathode") where y.val=0.0
extract name="Power" curve(v."anode", (v."anode"*i."cathode")) outf="perc_test.dat"
extract name="Pmax(W)" max(curve(v."anode", (v."anode"*i."cathode")))
extract name="Fill_factor" ($"Pmax(W)"/($"Isc(A)"*$"Voc(V)"))
extract name="Efficiency(%)" ($"Jsc(A/cm2)"*$"Voc(V)"*$"Fill_factor"*100/0.1)
```

이중반사방지막

3 3가지 기술을 적용한 Deckbuild Code

③ 구조



4 효율 비교

① Deckbuild를 활용한 효율 비교

```
ATLAS> log off
ATLAS>
ATLAS> tonyplot basic.log

ATLAS>
ATLAS> #parameter 추출
ATLAS> EXTRACT> init infile="basic.log"
EXTRACT> extract name="Isc(A)" max(i."cathode")
Isc(A)=1.27247e-007
EXTRACT> extract name="Jsc(A/cm2)" 1.27247e-007/4.2e-006
Jsc(A/cm2)=0.0302969
EXTRACT> extract name="Voc(V)" x.val from curve(v."anode",
Voc(V)=0.676785
EXTRACT> extract name="Power" curve(v."anode",
EXTRACT> extract name="Pmax(W)" max(curve(v."anode",
Pmax(W)=7.19014e-008
EXTRACT> extract name="Fill_factor"
Fill_factor=0.834909
EXTRACT> extract name="Efficiency(%)"
Efficiency(%)=17.1194
```

```
ATLAS> log off
ATLAS>
ATLAS> tonyplot solution.log

ATLAS>
ATLAS> ##parameter 추출
ATLAS> EXTRACT> init infile="solution.log"
EXTRACT> extract name="Isc(A)" max(i."cathode")
Isc(A)=1.40601e-007
EXTRACT> extract name="Jsc(A/cm2)" 1.40601e-007/4.2e-006
Jsc(A/cm2)=0.0334764
EXTRACT> extract name="Voc(V)" x.val from curve(v."anode", i."cathode") where y.val=0.0
Voc(V)=0.679865
EXTRACT> extract name="Power" curve(v."anode", (v."anode"*i."cathode")) outf="perc_test.dat"
EXTRACT> extract name="Pmax(W)" max(curve(v."anode", (v."anode"*i."cathode")))
Pmax(W)=7.97421e-008
EXTRACT> extract name="Fill_factor" (7.97421e-008/(1.40601e-007*0.679865))
Fill_factor=0.834212
EXTRACT> extract name="Efficiency(%)" (0.0334764*0.679865*0.834212*100/0.1)
Efficiency(%)=18.9862
```

효율 1.8668 증가

4 효율 비교

② 그래프를 이용한 효율 비교





사진 내용 출처 : 태양전지 효율 (tistory.com)
CFKO201230533383475.pdf (koreascience.or.kr)
PERC Solar Cells: What You Need to Know | EnergySage
산업부, AI기술 탑재된 반도체 상용화해 글로벌 명성 이어 나간다 - 인더스트리뉴스 (industrynews.co.kr)

감사합니다!